

NAG Library Routine Document

C05AYF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

C05AYF locates a simple zero of a continuous function in a given interval using Brent's method, which is a combination of nonlinear interpolation, linear extrapolation and bisection.

2 Specification

```
SUBROUTINE C05AYF (A, B, EPS, ETA, F, X, IUSER, RUSER, IFAIL)
INTEGER           IUSER(*), IFAIL
REAL (KIND=nag_wp) A, B, EPS, ETA, F, X, RUSER(*)
EXTERNAL          F
```

3 Description

C05AYF attempts to obtain an approximation to a simple zero of the function $f(x)$ given an initial interval $[a, b]$ such that $f(a) \times f(b) \leq 0$. The same core algorithm is used by C05AZF whose specification should be consulted for details of the method used.

The approximation x to the zero α is determined so that at least one of the following criteria is satisfied:

- (i) $|x - \alpha| \leq \text{EPS}$,
- (ii) $|f(x)| \leq \text{ETA}$.

4 References

Brent R P (1973) *Algorithms for Minimization Without Derivatives* Prentice–Hall

5 Arguments

- | | |
|--|--------------|
| 1: A – REAL (KIND=nag_wp) | <i>Input</i> |
| <i>On entry:</i> a , the lower bound of the interval. | |
| 2: B – REAL (KIND=nag_wp) | <i>Input</i> |
| <i>On entry:</i> b , the upper bound of the interval. | |
| <i>Constraint:</i> $B \neq A$. | |
| 3: EPS – REAL (KIND=nag_wp) | <i>Input</i> |
| <i>On entry:</i> the termination tolerance on x (see Section 3). | |
| <i>Constraint:</i> $\text{EPS} > 0.0$. | |
| 4: ETA – REAL (KIND=nag_wp) | <i>Input</i> |
| <i>On entry:</i> a value such that if $ f(x) \leq \text{ETA}$, x is accepted as the zero. ETA may be specified as 0.0 (see Section 7). | |

- 5: F – REAL (KIND=nag_wp) FUNCTION, supplied by the user. *External Procedure*
 F must evaluate the function f whose zero is to be determined.

The specification of F is:

```
FUNCTION F (X, IUSER, RUSER)
REAL (KIND=nag_wp) F
INTEGER           IUSER(*)
REAL (KIND=nag_wp) X, RUSER(*)
```

1: X – REAL (KIND=nag_wp) *Input*

On entry: the point at which the function must be evaluated.

2: IUSER(*) – INTEGER array *User Workspace*

3: RUSER(*) – REAL (KIND=nag_wp) array *User Workspace*

F is called with the arguments IUSER and RUSER as supplied to C05AYF. You should use the arrays IUSER and RUSER to supply information to F.

F must either be a module subprogram USED by, or declared as EXTERNAL in, the (sub) program from which C05AYF is called. Arguments denoted as *Input* must **not** be changed by this procedure.

- 6: X – REAL (KIND=nag_wp) *Output*

On exit: if IFAIL = 0 or 2, X is the final approximation to the zero. If IFAIL = 3, X is likely to be a pole of $f(x)$. Otherwise, X contains no useful information.

- 7: IUSER(*) – INTEGER array *User Workspace*
 8: RUSER(*) – REAL (KIND=nag_wp) array *User Workspace*

IUSER and RUSER are not used by C05AYF, but are passed directly to F and should be used to pass information to this routine.

- 9: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, A = $\langle\text{value}\rangle$ and B = $\langle\text{value}\rangle$.
 Constraint: A \neq B.

On entry, EPS = $\langle\text{value}\rangle$.
 Constraint: EPS > 0.0.

On entry, $F(A)$ and $F(B)$ have the same sign with neither equalling 0.0: $F(A) = \langle value \rangle$ and $F(B) = \langle value \rangle$.

IFAIL = 2

No further improvement in the solution is possible. EPS is too small: $\text{EPS} = \langle value \rangle$. The final value of X returned is an accurate approximation to the zero.

IFAIL = 3

The function values in the interval $[A, B]$ might contain a pole rather than a zero. Reducing EPS may help in distinguishing between a pole and a zero.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The levels of accuracy depend on the values of EPS and ETA. If full machine accuracy is required, they may be set very small, resulting in an exit with $\text{IFAIL} = 2$, although this may involve many more iterations than a lesser accuracy. You are recommended to set $\text{ETA} = 0.0$ and to use EPS to control the accuracy, unless you have considerable knowledge of the size of $f(x)$ for values of x near the zero.

8 Parallelism and Performance

C05AYF is not threaded in any implementation.

9 Further Comments

The time taken by C05AYF depends primarily on the time spent evaluating F (see Section 5).

If it is important to determine an interval of relative length less than $2 \times \text{EPS}$ containing the zero, or if F is expensive to evaluate and the number of calls to F is to be restricted, then use of C05AZF is recommended. Use of C05AZF is also recommended when the structure of the problem to be solved does not permit a simple F to be written: the reverse communication facilities of C05AZF are more flexible than the direct communication of F required by C05AYF.

10 Example

This example calculates an approximation to the zero of $e^{-x} - x$ within the interval $[0, 1]$ using a tolerance of $\text{EPS} = 1.0E-5$.

10.1 Program Text

```

! C05AYF Example Program Text
! Mark 26 Release. NAG Copyright 2016.

Module c05ayfe_mod

! C05AYF Example Program Module:
! Parameters and User-defined Routines

! .. Use Statements ..
Use nag_library, Only: nag_wp
! .. Implicit None Statement ..
Implicit None
! .. Accessibility Statements ..
Private
Public :: f
! .. Parameters ..
Integer, Parameter, Public :: nout = 6
Contains
Function f(x,iuser,ruser)

! .. Function Return Value ..
Real (Kind=nag_wp) :: f
! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: x
! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Inout) :: ruser(*)
Integer, Intent (Inout) :: iuser(*)
! .. Intrinsic Procedures ..
Intrinsic :: exp
! .. Executable Statements ..
f = exp(-x) - x

Return

End Function f
End Module c05ayfe_mod
Program c05ayfe

! C05AYF Example Main Program

! .. Use Statements ..
Use nag_library, Only: c05ayf, nag_wp
Use c05ayfe_mod, Only: f, nout
! .. Implicit None Statement ..
Implicit None
! .. Local Scalars ..
Real (Kind=nag_wp) :: a, b, eps, eta, x
Integer :: ifail
! .. Local Arrays ..
Real (Kind=nag_wp) :: ruser(1)
Integer :: iuser(1)
! .. Executable Statements ..
Write (nout,*) 'C05AYF Example Program Results'

a = 0.0E0_nag_wp
b = 1.0E0_nag_wp
eps = 1.0E-5_nag_wp
eta = 0.0E0_nag_wp

ifail = -1
Call c05ayf(a,b,eps,eta,f,x,iuser,ruser,ifail)

Write (nout,*)

Select Case (ifail)
Case (0)
    Write (nout,99999) 'Zero at X =', x
Case (2,3)

```

```
      Write (nout,99999) 'Final point = ', x
      End Select

99999 Format (1X,A,F12.5)
End Program c05ayfe
```

10.2 Program Data

None.

10.3 Program Results

```
C05AYF Example Program Results
Zero at X =      0.56714
```
